A2e. Checks of the definitions of $V_{\kappa,s}$ and $W_{\kappa,s}$

Checks for the case that $s = s_0 \in Z_{\geq 0}$ and κ such that W, respectively V, is not proportional to M. Determination of the leading term in the asymptotic behavior at 0.

Here the common factor $e^{-\tau/2}$ of the Whittaker functions can be ignored.

Terms of the expansion of $M_{\kappa,s}$

In[ ]:=
```
Clear[term, tau]
term[kp_, s_, n_] := tau ^ (s + 1 / 2 + n) Gamma[1 / 2 + s - kp + n]
   Gamma[1 / 2 + s - kp] ^ (- 1) Gamma[1 + 2 s + n] ^ (- 1) Gamma[1 + 2 s] Factorial[n] ^ (- 1)
```

W-Whittaker function

In[ ]:=
```
Clear[kp, s]
factWp = Gamma[- 2 s] Gamma[1 / 2 - s - kp] ^ (- 1)
factWm = Gamma[2 s] Gamma[1 / 2 + s - kp] ^ (- 1)
Clear[gamsub]
gamsub[xx_] := Gamma[xx] → Pi Gamma[1 - xx] ^ (- 1) Sin[Pi xx] ^ (- 1)
```

Out[ ]= $\frac{\text{Gamma}[-2\ s]}{\text{Gamma}\left[\frac{1}{2} - \text{kp} - s\right]}$

Out[ ]= $\frac{\text{Gamma}[2\ s]}{\text{Gamma}\left[\frac{1}{2} - \text{kp} + s\right]}$

For W we work with the assumption that $s_0 - \kappa = p_0 - 1/2$ with a positive integer p_0.

First terms

The terms with order $0 \leq m < 2\, s_0$ should give holomorphic conditions in s by themselves.

In[ ]:=
```
Clear[eta]
factWm term[kp, -s, m] //.
 {gamsub[1 + m - 2 s], gamsub[1 - 2 s], Sin[π (1 + m - 2 s)] → (-1) ^ m Sin[π (1 - 2 s)]}

wlw = % /. s → s0 + eta //. {Sin[π (1 + m - 2 (eta + s0))] → Sin[-2 Pi eta] (-1) ^ (1 + m - 2 s0),
      s0 - kp → p0 - 1 / 2, -kp - s0 → p0 - 1 / 2 - 2 s0} // Simplify ;
Series[%, {eta, 0, 0}]
% /. m → 0
```

Out[ ]= $\dfrac{(-1)^m \, tau^{\frac{1}{2}+m-s} \, Gamma\left[\frac{1}{2} - kp + m - s\right] Gamma[-m + 2 s]}{m! \, Gamma\left[\frac{1}{2} - kp - s\right] Gamma\left[\frac{1}{2} - kp + s\right]}$

Out[ ]= $\dfrac{(-1)^m \, tau^{\frac{1}{2}+m-s0} \, Gamma[m + p0 - 2 s0] \, Gamma[-m + 2 s0]}{m! \, Gamma[p0] \, Gamma[p0 - 2 s0]} + O[eta]^1$

Out[ ]= $\dfrac{tau^{\frac{1}{2}-s0} \, Gamma[2 s0]}{Gamma[p0]} + O[eta]^1$

Holomorphic in s at s_0.

Non-zero starting term with factor $tau^{\frac{1}{2}-s0}$

Case s = 0

In[ ]:=
```
factWm term[kp, -s, m] + factWp term[kp, s, m] //. {kp → 1 / 2 - p0,
    Gamma[-2 s] → Gamma[1 - 2 s] / (-2 s), Gamma[2 s] → Gamma[1 + 2 s] / (2 s)} // Simplify
Series[
 %,
 {s,
  0,
  0}]
```

Out[ ]= $-\left(\left(tau^{\frac{1}{2}+m-s} \, Gamma[1 - 2 s] \, Gamma[1 + 2 s] \left(tau^{2 s} \, Gamma[1 + m - 2 s] \, Gamma[m + p0 + s] - Gamma[m + p0 - s] \, Gamma[1 + m + 2 s]\right)\right) \Big/ \left(2 s \, m! \, Gamma[1 + m - 2 s] \, Gamma[p0 - s] \, Gamma[p0 + s] \, Gamma[1 + m + 2 s]\right)\right)$

Out[ ]= $-\left(\left(tau^{\frac{1}{2}+m} \, Gamma[m + p0] \, (Log[tau] - 2 \, PolyGamma[0, 1 + m] + PolyGamma[0, m + p0])\right) \Big/ \left(m! \, Gamma[1 + m] \, Gamma[p0]^2\right)\right) + O[s]^1$

Holomorphic, with a logarithmic term .

Case 0 < s

In[ ]:=
```
factWm term[kp, -s, m + 2 s0] + factWp term[kp, s, m] /. Gamma[1 - 2 s] → Gamma[-2 s] (-2 s) //.
        {kp → 1 / 2 - p0} /. s → s0 + eta /. gamsub[-2 (eta + s0)]  //.
    {Csc[2 π (eta + s0)] → Csc[2 Pi eta] (-1) ^ (2 s0)} // Simplify;
Series[%, {eta, 0, 0}] // Simplify
% /. m → 0 // Simplify
```

Out[ ]=
```
((-1)^(2 s0) tau^(1/2+m+s0) Gamma[m + p0 + s0] (-(m + 2 s0)! Gamma[1 + m] Gamma[1 + 2 s0]
        (-2 PolyGamma[0, 1 + m] + PolyGamma[0, m + p0 + s0] + 2 (Log[tau] + PolyGamma[0, 1 + 2 s0])) +
      2 m! Gamma[2 s0] Gamma[1 + m + 2 s0] (1 + 2 s0 PolyGamma[0, 2 s0] -
          s0 PolyGamma[0, m + p0 + s0] + 2 s0 PolyGamma[0, 1 + m + 2 s0]))) /
  (2 m! (m + 2 s0)! Gamma[1 + m] Gamma[p0 - s0] Gamma[p0 + s0] Gamma[1 + 2 s0] Gamma[1 + m + 2 s0]) +
 O[eta]^1
```

Out[ ]=
```
((-1)^(1+2 s0) tau^(1/2+s0) (-2 Gamma[2 s0]
        (1 + 2 s0 PolyGamma[0, 2 s0] - s0 PolyGamma[0, p0 + s0] + 2 s0 PolyGamma[0, 1 + 2 s0]) +
      (2 s0)! (PolyGamma[0, p0 + s0] + 2 (EulerGamma + Log[tau] + PolyGamma[0, 1 + 2 s0])))) /
  (2 × (2 s0)! Gamma[p0 - s0] Gamma[1 + 2 s0]) + O[eta]^1
```

Holomorphic, zero if $p_0 < s_0$.

So $W_{\kappa,s}$ is holomorphic and even in s.

If $s_0 > 0$ it starts with $\tau^{-s_0+1/2}$.

If $s_0 = 0$ its main contribution at zero is logarithmic .

V-Whittaker function

In[ ]:=
```
factVp = I E ^ (Pi I s) (Pi / Sin[2 Pi s]) Gamma[1 / 2 - s + kp] ^ (-1) Gamma[1 + 2 s] ^ (-1)
factVm = -I E ^ (-Pi I s) (Pi / Sin[2 Pi s]) Gamma[1 / 2 + s + kp] ^ (-1) Gamma[1 - 2 s] ^ (-1)
```

Out[ ]=
$$\frac{i\, e^{i \pi s} \pi \,\mathrm{Csc}[2 \pi s]}{\mathrm{Gamma}\left[\frac{1}{2} + kp - s\right] \mathrm{Gamma}[1 + 2 s]}$$

Out[ ]=
$$-\frac{i\, e^{-i \pi s} \pi \,\mathrm{Csc}[2 \pi s]}{\mathrm{Gamma}[1 - 2 s]\, \mathrm{Gamma}\left[\frac{1}{2} + kp + s\right]}$$

For V with $s_0 + \kappa = q_0 - 1/2$

First terms

Take 0 <= m < 2 s_0

In[ ]:=
```
Clear[eta]
vlw = factVm term[kp, -s, m] /. kp → -s0 + q0 - 1 / 2 /. s → eta + s0 /. gamsub[1 + m - 2 (eta + s0)] /.
    gamsub[eta + q0] //. {Csc[2 π (eta + s0)] → (-1) ^ (2 s0) Csc[2 Pi eta],
    Sin[π (1 + m - 2 (eta + s0))] → Sin[2 Pi eta] (-1) ^ (m - 2 s0),
    Sin[π (eta + q0)] → Sin[Pi eta] (-1) ^ q0} // Simplify
```

Out[ ]= $-\frac{1}{\pi\, m!} i\, (-1)^{m+q0}\, e^{-i\,\pi\,(eta+s0)}\, tau^{\frac{1}{2}-eta+m-s0}\, \text{Gamma}[1-eta+m-q0]\, \text{Gamma}[-m+2\,(eta+s0)]\, \text{Sin}[eta\,\pi]$

Suppose $m \geq q_0$

In[ ]:=
```
Series[vlw, {eta, 0, 0}]
```

Out[ ]= $O[eta]^1$

The coefficients with m in $[q_0, 2\, s_0)$ are zero

Suppose $m < q_0$

In[ ]:=
```
(vlw /. gamsub[1 - eta + m - q0] //. {Csc[π (1 - eta + m - q0)] → Csc[Pi eta] (-1) ^ (m - q0)} //
    Simplify) /. (-1)^(2 m) → 1;
Series[%, {eta, 0, 0}] // Simplify
% /. m → 0
```

Out[ ]= $-\frac{i\, e^{-i\,\pi\, s0}\, tau^{\frac{1}{2}+m-s0}\, \text{Gamma}[-m+2\, s0]}{m!\, \text{Gamma}[-m+q0]} + O[eta]^1$

Out[ ]= $-\frac{i\, e^{-i\,\pi\, s0}\, tau^{\frac{1}{2}-s0}\, \text{Gamma}[2\, s0]}{\text{Gamma}[q0]} + O[eta]^1$

For $0 \leq m < \min(q_0, 2\, s_0)$

the terms are holomorphic at s_0.

The initial term is non-zero, with $tau^{\frac{1}{2}-s0}$

Case $s_0 = 0$

In[ ]:=
```
w0 = factVm term[kp, -s, m] + factVp term[kp, s, m] /. kp → q0 - 1 / 2 /. gamsub[1 + m - q0 - s] /.
      gamsub[1 - q0 + s] /. gamsub[1 + m - q0 + s] /. gamsub[1 - q0 - s] //.
   {Csc[π (1 + m - q0 - s)] → (-1) ^ (m - q0) Csc[Pi s], Sin[π (1 - q0 - s)] → Sin[Pi s] (-1) ^ q0,
    Csc[π (1 + m - q0 + s)] → Csc[Pi s] (-1) ^ (1 + m - q0),
    Sin[π (1 - q0 + s)] → Sin[Pi s] (-1) ^ (1 - q0)} // Simplify ;
Series[%, {s, 0, 0}] // Simplify ;
% /. (-1) ^ (2 q0) → 1 /. (-1) ^ (-2 q0) → 1
% /. m → 0 // Simplify
```

Out[ ]= $\left((-1)^{1+m-2\,q0}\,\text{tau}^{\frac{1}{2}+m}\,(\pi - i\,\text{Log[tau]} + 2\,i\,\text{PolyGamma}[0,\,1+m] - i\,\text{PolyGamma}[0,\,-m+q0])\right) / (m!\,\text{Gamma}[1+m]\,\text{Gamma}[-m+q0]) + O[s]^1$

Out[ ]= $\frac{1}{\text{Gamma}[q0]}\,i\,(-1)^{-2\,q0}\,\sqrt{\text{tau}}\,(2\,\text{EulerGamma} + i\,\pi + \text{Log[tau]} + \text{PolyGamma}[0,\,q0]) + O[s]^1$

If s = 0 all terms are holomorphic at s=0. The initial term contains a non-zero multiple of $\tau^{1/2}\log(\tau)$

Terms with $m \geq 2\,s_0 > 0$

In[ ]:=
```
vhg = Simplify[
    factVm term[kp, -s, m] + factVp term[kp, s, m] /. kp → -s0 + q0 - 1 / 2 /. s → eta + s0 /.
      gamsub[1 + m - 2 (eta + s0)] /. gamsub[-eta + q0 - 2 s0] /. gamsub[1 - eta - q0]] //.
  {Csc[2 π (eta + s0)] → Csc[2 Pi eta] (-1) ^ (2 s0), Sin[π (1 + m - 2 (eta + s0))] →
    -Sin[2 Pi eta] (-1) ^ (m - 2 s0), Sin[π (-eta + q0 - 2 s0)] → -Sin[Pi eta] (-1) ^ (q0 - 2 s0),
    Sin[π (-1 + eta + q0)] → -Sin[Pi eta] (-1) ^ q0} // Simplify
```

Out[ ]= $-\left(\left(i\,(-1)^{q0}\,e^{-i\,\pi\,(\text{eta}+s0)}\,\text{tau}^{\frac{1}{2}-\text{eta}+m-s0}\,\left(e^{2\,i\,\pi\,(\text{eta}+s0)}\,\pi\,\text{tau}^{2\,(\text{eta}+s0)}\,\text{Csc}[2\,\text{eta}\,\pi]\,\text{Gamma}[1+\text{eta}+m-q0+2\,s0] + (-1)^{1+m}\,\text{Gamma}[1-\text{eta}+m-q0]\,\text{Gamma}[-m+2\,(\text{eta}+s0)]\,\text{Gamma}[1+m+2\,(\text{eta}+s0)]\right)\,\text{Sin}[\text{eta}\,\pi]\right) / (\pi\,m!\,\text{Gamma}[1+m+2\,(\text{eta}+s0)])\right)$

Case $1 \leq q_0 \leq 2\,s0$

In[ ]:=
```
Series[vhg, {eta, 0, 0}] //. {Gamma[1 + xx_] :→ Factorial[xx]}
```

Out[ ]= $-\frac{i\,(-1)^{q0}\,e^{i\,\pi\,s0}\,\text{tau}^{\frac{1}{2}+m+s0}\,(m-q0+2\,s0)!}{2\,m!\,(m+2\,s0)!} + O[\text{eta}]^1$

Holomorphic, with non-zero value

Case $2\,s0 < q_0 \leq m$

In[ ]:=
```
vhg /. gamsub[-m + 2 (eta + s0)] /. gamsub[1 + m - 2 (eta + s0)] //.
   {Csc[π (-m + 2 (eta + s0))] → Csc[2 Pi eta] (-1) ^ (-m + 2 s0),
    Sin[π (1 + m - 2 (eta + s0))] → Sin[2 Pi eta] (-1) ^ (m - 2 s0)} // Simplify ;
Series[%, {eta, 0, 0}] //. Gamma[1 + xx_] :→ Factorial[xx] // Simplify
```

Out[ ]= $-\frac{i\,(-1)^{q0}\,e^{i\,\pi\,s0}\,\text{tau}^{\frac{1}{2}+m+s0}\,(m-q0+2\,s0)!}{2\,m!\,(m+2\,s0)!} + O[\text{eta}]^1$

Holomorphic and non-zero.

Case $2\, s0 \le m < q_0$

In[ ]:=
```
vhg /. gamsub[1 - eta + m - q0] //. {Csc[π (1 - eta + m - q0)] → Csc[Pi eta] (-1)^(m - q0)} // Simplify ;
(Series[%, {eta, 0, 0}] /. Gamma[xx_] :> Factorial[xx - 1] // Simplify) /. (-1)^(2 m) → 1
```

Out[ ]=
$$\frac{1}{2\, m!} i\, e^{-i \pi\, s0}\, tau^{\frac{1}{2}+m-s0} \left(\frac{2 \times (-1 - m + 2\, s0)!}{(-1 - m + q0)!} + \frac{(-1)^{1+q0}\, e^{2 i \pi\, s0}\, tau^{2\, s0}\, (m - q0 + 2\, s0)!}{(m + 2\, s0)!} \right) + O[eta]^1$$

This is also holomorphic .

The function $V_{\kappa,s}(\tau)$ is holomorphic in s at all points with Re(s)≥0.

At s=0 the main term at 0 is a non-zero multiple of $\tau^{1/2} \log\tau$

At points $s_0 \in (1/2)\, Z_{\ge 1}$

the expansion starts with a multiple of $\tau^{1/2 - s_0}$.